
1. 調査概要 
 

Microsft社のWebブラウザ「Microsoft Internet Explorer」上では、JScript(JavaScript互

換)だけではなく、VBScript も動作する。 

汚染データを、VBScript上の文字列リテラルとして用いる場合のサニタイジング法(エスケープ法)

について検討した結果をここに記す。 

 

サニタイジング処理の対象文字は、以下の二つが基本である。 

 「"(ダブルクォート)」 

 改行 (Cr 及び Lf) 

 

HTML内に VBScript を記述する場合には、以下の三つの文字もサニタイジング対象となる。 

 「/(スラッシュ)」 

 「<(小なり記号)」 

 「>(大なり記号)」 

これらは、VBScript中に「</script>」などという文字列を挿入されて、HTMLパーサーが誤

動作しないために必要である。 

 

念のために、以下の文字もサニタイジングの対象としてもよい。 

 「’(シングルクォート)」 

VBScriptでは、文字列リテラルは「"(ダブルクォート)」で囲むため、「'(シングルクォート)」のサ

ニタイジング処理は必要ないと思われるが、念のためにサニタイジング処理してもよいだろう。 

 

 

2. HTML上の JavaScriptのエスケープ法 
 

VBScriptについて検討する前に、HTML上の JavaScriptにおいて、汚染データを文字列リテ

ラルとして使う場合について簡単に検討する。 

 

HTML上の JavaScript では、以下の文字がサニタイジング処理の対象となっている。 

 

 「"(ダブルクォート)」 

 「'(シングルクォート)」 

 「\(バックスラッシュ)」 

 「改行コード(Cr 及び Lf)」 

 「/(スラッシュ)」 

 「<(小なり記号)」 

 「>(大なり記号)」 

 

以上の検討内容を、VBScriptに対して適用していく。 

3. HTML上の VBScriptのエスケープ法 
 

3.1. VBScript での文字列リテラルに対してのサニタイジング処理の

基本 
 

VBScriptでは、文字列リテラルを「"(ダブルクォート)」で囲む。また、文字列リテラル上の「"(ダブル

クォート)」は、「""(ダブルクォート二個)」に置換(エスケープ)することで、文字列リテラルとしての(デ

ータとしての)「"(ダブルクォート)」を表現することができる。 

 

VBScriptでは、コードは改行で区切られて記述される。文字列のデータとして改行を示す場合、

「vbLf」「vbCr」「vbCrLf」などの定数が用意されているため、それらに置換する。 

 

最後に文字列データの連結には「&」を用いる。 

 

 

3.2. HTML 上の VBScript の場合の注意点 
 

HTML上のスクリプト・コードは、基本的には以下の手順でスクリプト・実行エンジンに渡される 

1. HTTPレスポンス・メッセージをWebブラウザが受信 

2. HTTPレスポンス・メッセージのボディ部分が、HTMLレンダリング・エンジンに渡される 

3. HTTPレンダリング・エンジンは、「<script>」から「</script>」までを抜き出し、その部分をス

クリプト・実行エンジンに渡す 

 

以上の流れを考えれば、VBScriptコード中に「</script>」などのデータが挿入された場合、3の

処理が誤動作する可能性がある。 

よって、HTML上のスクリプト・コードでは「/」「<」「>」をサニタイジング処理しておく必要がある。 

 

VBScriptの場合、Chr() 関数を使い、以下の三つの文字を関数に置き換えることでサニタイジン

グ処理することができる。 

 「/(スラッシュ)」 → Chr(47) 

 「<(小なり記号)」 → Chr(60) 

 「>(大なり記号)」 → Chr(62) 

 

 

3.3. 「'(シングルクォート)」の扱い 
 

VBScriptでは「'(シングルクォート)」は、コメントの開始を意味する。(JavaScriptでいえば「//」と同

義と考えてよい) 

よって、VBScriptに関しては、「'(シングルクォート)」をサニタイジング処理する必要はないといえる

が、Chr() 関数を使って以下のように置き換えてもよい。 

 

 「'(シングルクォート)」 → Chr(39) 

3.4. まとめ 
 

HTML上の VBScriptに対して、汚染データを文字列リテラルの一部として埋め込む場合、以下

の文字をサニタイジング処理する。 

 

 「"(ダブルクォート)」 → 「""」 

 CrLf(改行)  → vbCrLf 

 Cr(改行)   → vbCrLf 

 Lf(改行)   → vbLf 

 「/(スラッシュ)」 → Chr(47) 

 「<(小なり記号)」 → Chr(60) 

 「>(大なり記号)」 → Chr(62) 

 「'(シングルクォート)」→ Chr(39) 

(念のため、「'(シングルクォート)」)も含めた 

 

 

4. 検証結果 
 

4.1. 検証環境 
 

以下の環境で、検証を行った。 

 Microsoft WindowsXP 日本語版 SP3 上の IIS5.1 + ASP 

 Microsoft WindowsXP 日本語版 SP3 上のMicrosoft Internet Explorer 6 SP3 

4.2. 検証コード(IIS+ASP) 
 
<% 

 Option Explicit 

 Dim myArg1 

 myArg1 = "" 

 myArg1 = Request.Form("arg1") 

 myArg1 = VBEscape(myArg1,True,True) 

%> 

<HTML> 

 <HEAD><TITLE>VBEscape テスト</TITLE></HEAD> 

 <BODY> 

  <CENTER> 

   <FORM ACTION="" METHOD="post"> 

    <TEXTAREA NAME="arg1" ROWS="10" COLS="72"><% = 

Server.HTMLEncode(Request.Form("arg1")) %></TEXTAREA><BR> 

    <INPUT TYPE="submit"> 

   </FORM> 

   <HR> 

   <% If 0 < Len(myArg1) Then %> 

    <SCRIPT LANGUAGE="VBScript"> 

     MsgBox(<% = myArg1 %>) 

    </SCRIPT> 

   <% End If%> 

  </CENTER> 

 </BODY> 

</HTML> 

<% 

Function VBEscape(iStr, iFlg1, iFlg2) 

  Dim myStr 

  Dim chkStr 

 Rem 「"」→「\"」, 「\」→「\\」にして退避する 

  myStr = Replace(iStr,"\","\\") 

  myStr = Replace(myStr,"""","\""") 

 Rem Cr, Lf をエスケープ 

  myStr = Replace(myStr,vbCrLf,""" & vbCrLf & """) 

  myStr = Replace(myStr,vbLf,""" & vbLf & """) 

  myStr = Replace(myStr,vbCr,""" & vbCr & """) 

 Rem 「/」「<」「>」をエスケープ 

  If iFlg1 = True Then 

   myStr = Replace(myStr,"/",""" & Chr(47) & """) 

   myStr = Replace(myStr,"<",""" & Chr(60) & """) 

   myStr = Replace(myStr,">",""" & Chr(62) & """) 

  End If 

 Rem 「'」をエスケープ 

  If iFlg2 = True Then 

   myStr = Replace(myStr,"'",""" & Chr(39) & """) 

  End If 

  myStr = """" & myStr & """" 

 Rem 空文字列を削除し、最適化する 

  myStr = Replace(myStr," & """" & "," & ") 

 Rem 先頭に空文字があるなら削除 

  chkStr = """"" & " 

  If 1 = InStr(1,myStr,chkStr) Then 

   myStr = Mid(myStr,Len(chkStr)+1) 

  End If 

 Rem 後尾に空文字があるなら削除 

  chkStr = " & """"" 

  If Len(chkStr) < Len(myStr) Then 

   If chkStr = Right(myStr,Len(chkStr)) Then 

    myStr = Left(myStr,Len(myStr) - Len(chkStr)) 

   End If 

  End If 

  myStr = TRim(myStr) 

 Rem 退避していた「"」を戻す 

  myStr = Replace(myStr,"\\","\") 

  myStr = Replace(myStr,"\""","""""") 

  VBEscape = myStr 

 End Function%> 

図 4.2-1 : VBScript のサニタイジング処理が正しいかどうか検証するための IIS+ASP のコード 

 

IIS+ASPのコードは、図 4.2-1である。 

このコードの VBEscape(入力データ、フラグ 1、フラグ 2)が、エスケープ関数である。 

(フラグ 1=Trueの時、「/(スラッシュ)」「<(小なり記号)」「>(大なり記号)」らもエスケープする) 

(フラグ 2=Trueの時、「'(シングルクォート)」もエスケープする) 

(空文字列の結合処理を削除するため、コードが若干複雑になっているが、最終的には、3.4で一

覧にした文字全てをエスケープしていればよいので、図 4.2-2のように簡略化してもよい) 

 

Function VBEscape(iStr, iFlg1, iFlg2) 

  Dim myStr 

 Rem 「"」→「""」 

  myStr = Replace(iStr,"""","""""") 

 Rem Cr, Lf をエスケープ 

  myStr = Replace(myStr,vbCrLf,""" & vbCrLf & """) 

  myStr = Replace(myStr,vbLf,""" & vbLf & """) 

  myStr = Replace(myStr,vbCr,""" & vbCr & """) 

 Rem 「/」「<」「>」をエスケープ 

  If iFlg1 = True Then 

   myStr = Replace(myStr,"/",""" & Chr(47) & """) 

   myStr = Replace(myStr,"<",""" & Chr(60) & """) 

   myStr = Replace(myStr,">",""" & Chr(62) & """) 

  End If 

 Rem 「'」をエスケープ 

  If iFlg2 = True Then 

   myStr = Replace(myStr,"'",""" & Chr(39) & """) 

  End If 

  VBEscape = """" & myStr & """" 

End Function 

図 4.2-2 : 簡略化したエスケープ処理関数 

4.3. 検証結果 
 

以下の図より、適切にエスケープされていることを確認した。 

 

 
図 4.3-1 : 図 4.2-1 にアクセスした結果 

 

 
図 4.3-2 : 図 4.3-1 の HTML ソース 

5. 検証作業者 
 

NTTコミュニケーションズ株式会社 

ITマネジメントサービス事業部ネットワークマネジメントサービス部 

セキュリティオペレーションセンター 

佐名木 智貴 

 

 

6. 参考 
 

 メーリングリスト「Sea Surfers ML」 

http://www.freeml.com/seasurfers 

 T.Teradaの日記 

http://d.hatena.ne.jp/teracc/ 

 

 

7. 履歴 
 

 2009年 05月 26日 ： ver1.0 最初の公開 

 2009年 09月 03日 ： ver1.1 「4.2 検証コード(IIS+ASP)」のサンプルコードのバグ修正 

     (http://d.hatena.ne.jp/teracc/20090601#1243809381)。 

     「6 参考」にWebページを一つ追加。 

 

 

8. 最新版の公開 URL 
 

http://www.ntt.com/icto/security/data/soc.html#security_report 

 

 

9. 本レポートに関する問合せ先 
 

NTTコミュニケーションズ株式会社 

ITマネジメントサービス事業部ネットワークマネジメントサービス部 

セキュリティオペレーションセンター 

 

e-mail:  scan@ntt.com 

 

 

以 上 


